# Disjunctive Delimited Control 

Alexander Vandenbroucke ${ }^{1}$ and Tom Schrijvers ${ }^{2}$<br>${ }^{1}$ Standard Chartered alexander.vandenbroucke@sc.com<br>${ }^{2}$ KU Leuven tom.schrijvers@kuleuven.be


#### Abstract

Delimited control is a powerful mechanism for programming language extension which has been recently proposed for Prolog (and implemented in SWI-Prolog). By manipulating the control flow of a program from inside the language, it enables the implementation of powerful features, such as tabling, without modifying the internals of the Prolog engine. However, its current formulation is inadequate: it does not capture Prolog's unique non-deterministic nature which allows multiple ways to satisfy a goal. This paper fully embraces Prolog's non-determinism with a novel interface for disjunctive delimited control, which gives the programmer not only control over the sequential (conjunctive) control flow, but also over the non-deterministic control flow. We provide a meta-interpreter that conservatively extends Prolog with delimited control and show that it enables a range of typical Prolog features and extensions, now at the library level: findall, cut, branch-and-bound optimisation, probabilistic programming, ...


Keywords: delimited control, disjunctions, Prolog, meta-interpreter, branch-and-bound

## 1 Introduction

Delimited control is a powerful programming language mechanism for control flow manipulation that was developed in the late '80s in the context of functional programming [512]. Schrijvers et al. [10] have recently ported this mechanism to Prolog.

This port enables powerful applications in Prolog, such as high-level implementations of both tabling [3] and algebraic effects \& handlers 7. Yet, at the same time, their work leaves much untapped potential, as it fails to recognise the unique nature of Prolog when compared to functional and imperative languages that have previously adopted delimited control.

Indeed, computations in other languages have only one continuation, i.e., one way to proceed from the current point to a result. In contrast, at any point in a Prolog continuation, there may be multiple ways to proceed and obtain a result. More specifically, we can distinguish 1) the success or conjunctive continuation which proceeds with the current state of the continuation; and 2) the failure or
disjunctive continuation which bundles the alternative ways to proceed, e.g., if the conjunctive continuation fails.

The original delimited control only accounts for one continuation, which Schrijvers et al. have unified with Prolog's conjunctive continuation. More specifically, for a given subcomputation, they allow to wrest the current conjunctive continuation from its track, and to resume it at leisure, however many times as desired. Yet, this entirely ignores the disjunctive continuation, which remains as and where it is.

In this work, we adapt delimited control to embrace the whole of Prolog and capture both the conjunctive and the disjunctive continuations. This makes it possible to manipulate Prolog's built-in search for custom search strategies and enables clean implementations of, e.g., findall/3 and branch-and-bound. This new version of delimited control has an executable specification in the form of a meta-interpreter (Section 3), that can run both the above examples, amongst others.

## 2 Overview and Motivation

### 2.1 Background: Conjunctive Delimited Control

In earlier work, Schrijvers et al. 10 have introduced a Prolog-compatible interface for delimited control that consists of two predicates: reset/3 and shift/1. To paraphrase their original description, reset (Goal, ShiftTerm, Cont) executes Goal, and, 1. if Goal fails, reset/3 also fails; 2. if Goal succeeds, then reset/3 also succeeds and unifies Cont and ShiftTerm with 0; 3. if Goal calls shift(Term), then the execution of Goal is suspended and reset/3 succeeds immediately, unifying ShiftTerm with Term and Cont with the remainder of Goal. The shift/reset pair resembles the more familiar catch/throw predicates, with the following differences: shift/1 does not copy its argument (i.e., it does not refresh the variables), it does not delete choice points, and also communicates the remainder of Goal to reset/3.

Obliviousness to Disjunctions This form of delimited control only captures the conjunctive continuation. For instance reset ( (shift (a), G1) , Term, Cont) captures in Cont goal G1 that appears in conjunction to shift(a). In a low-level operational sense this corresponds to delimited control in other (imperative and functional) languages where the only possible continuation to capture is the computation that comes sequentially after the shift. Thus this approach is very useful for enabling conventional applications of delimited control in Prolog.

In functional and imperative languages delimited control can also be characterised at a more conceptual level as capturing the entire remainder of a computation. Indeed, in those languages the sequential continuation coincides with the entire remainder of a computation. Yet, the existing Prolog approach fails to capture the entire remainder of a goal, as it only captures the conjunctive continuation and ignores any disjunctions. This can be illustrated by the reset ( (shift (a) , G1; G2) , Term, Cont) which only captures the conjunctive
continuation G1 in Cont and not the disjunctive continuation G2. In other words, only the conjunctive part of the goal's remainder is captured.

This is a pity because disjunctions are a key feature of Prolog and many advanced manipulations of Prolog's control flow involve manipulating those disjunctions in one way or another.

### 2.2 Delimited Continuations with Disjunction

This paper presents an approach to delimited control for Prolog that is in line with the conceptual view that the whole remainder of a goal should be captured, including in particular the disjunctive continuation.

For this purpose we modify the reset/3 interface, where depending on Goal, reset (Pattern, Goal, Result) has three possible outcomes:

1. If Goal fails, then the reset succeeds and unifies Result with failure. For instance,
```
?- reset(_,fail,Result).
Result = failure.
```

2. If Goal succeeds, then Result is unified with success (PatternCopy, DisjCont) and the reset succeeds. Here DisjCont is a goal that represents the disjunctive remainder of Goal. For instance,
```
?- reset(X,(X = a; X = b),Result).
X = a, Result = success(Y,Y = b).
```

Observe that, similar to findall/3, the logical variables in DisjCont have been renamed apart to avoid interference between the branches of the computation. To be able to identify any variables of interest after renaming, we provide PatternCopy as a likewise renamed-apart copy of Pattern.
3. If Goal calls shift(Term), then the reset succeeds and Result is unified with shift(Term, ConjCont, PatternCopy, DisjCont). This contains in addition to the disjunctive continuation also the conjunctive continuation. The latter is not renamed apart and can share variables with Pattern and Term. For instance,

```
?- reset(X,(shift(t),X = a; X = b),Result).
Result = shift(t,X = a, Y, Y = b).
```

Note that reset ( $\mathrm{P}, \mathrm{G}, \mathrm{R}$ ) always succeeds if $R$ is unbound and never leaves choicepoints.

Encoding findall/3 Section 4 presents a few larger applications, but our encoding of findall/3 with disjunctive delimited control already gives some idea of the expressive power:

```
findall(Pattern,Goal,List) :-
    reset(Pattern,Goal,Result),
```

```
    findall_result(Result,Pattern,List).
findall_result(failure,_, []).
findall_result(success(PatternCopy,DisjCont),Pattern,List) :-
    List = [Pattern|Tail],
    findall(PatternCopy,DisjCont,Tail).
```

This encoding is structured around a reset/3 call of the given Goal followed by a case analysis of the result. Here we assume that shift/1 is not called in Goal, which is a reasonable assumption for plain findall/3.

Encoding !/O Our encoding of the !/0 operator illustrates the use of shift/1:

```
cut :- shift(cut).
scope(Goal) :-
    copy_term(Goal,Copy),
    reset(Copy,Copy,Result),
    scope_result(Result,Goal,Copy).
scope_result(failure,_,_) :-
    fail.
scope_result(success(DisjCopy,DisjGoal),Goal,Copy) :-
    Goal = Copy.
scope_result(success(DisjCopy,DisjGoal),Goal,Copy) :-
    DisjCopy = Goal,
    scope(DisjGoal).
scope_result(shift(cut,ConjGoal,DisjCopy,DisjGoal),Goal,Copy) :-
    Copy = Goal,
    scope(ConjGoal).
```

The encoding provides cut/0 as a substitute for !/0. Where the scope of regular cut is determined lexically, we use scope/1 here to define it dynamically. For instance, we encode

```
p(X,Y) :- q(X), !, r(Y).
p(4,2).
p(X,Y) :- scope(p_aux (X,Y)).
as p_aux(X,Y) :- q(X), cut, r(Y).
p_aux(4,2).
```

The logic of cut is captured in the definition of scope/1; all the cut/0 predicate does is request the execution of a cut with shift/1.

In scope/1, the Goal is copied to avoid instantiation by any of the branches. The copied goal is executed inside a reset/3 with the copied goal itself as the pattern. The scope_result/3 predicate handles the result:

- failure propagates with fail;
- success creates a disjunction to either unify the initial goal with the now instantiated copy to propagate bindings, or to invoke the disjunctive continuation;
- shift (cut) discards the disjunctive continuation and proceeds with the conjunctive continuation only.


## 3 Meta-Interpreter Semantics

We provide an accessible definition of disjunctive delimited control in the form of a meta-interpreter. Broadly speaking, it consists of two parts: the core interpreter, and a top level predicate to initialise the core and interpret the results.

### 3.1 Core Interpreter

Figure 1 defines the interpreter's core predicate, eval(Conj, PatIn, Disj, PatOut, Result). It captures the behaviour of reset (Pattern, Goal, Result) where the goal is given in the form of a list of goals, Conj, together with the alternative branches, Disj. The latter is renamed apart from Conj to avoid conflicting instantiations.

The pattern that identifies the variables of interest (similar to findall/3) is present in three forms. Firstly, PatIn is an input argument that shares the variables of interest with Conj (but not with Disj). Secondly, PatOut outputs the instantiated pattern when the goal succeeds or suspends on a shift/1. Thirdly, the alternative branches Disj are of the form alt (BranchPatIn, BranchGoal) with their own copy of the pattern.

When the conjunction is empty (1-4), the output pattern is unified with the input pattern, and success/2 is populated with the information from the alternative branches.

When the first conjunct is true/0 (5-6), it is dropped and the meta-interpreter proceeds with the remainder of the conjunction. When it is a composite conjunction $(G 1, G 2)(7-8)$, the individual components are added separately to the list of conjunctions.

When the first conjunct is fail/0 (9-10), the meta-interpreter backtracks explicitly by means of auxiliary predicate backtrack/3.

```
backtrack(Disj,PatOut,Result) :-
    ( empty_alt(Disj) ->
        Result = failure
    ; Disj = alt(BranchPatIn,BranchGoal) ->
        empty_alt(EmptyDisj),
        eval([BranchGoal],BranchPatIn,EmptyDisj,PatOut,Result)
    ).
empty_alt(alt(_,fail)).
```

```
eval([],PatIn,Disj,PatOut,Result) :- !,
    PatOut = PatIn,
    Disj = alt(BranchPatIn,BranchGoal),
    Result = success(BranchPatIn,BranchGoal).
eval([true|Conj],PatIn,Disj,PatOut,Result) :- !,
    eval(Conj,PatIn,Disj,PatOut,Result).
eval([(G1,G2)|Conj],PatIn,Disj,PatOut,Result) :- !,
    eval([G1,G2|Conj],PatIn,Disj,PatOut,Result).
eval([fail|_Conj],_,Disj,PatOut,Result) :- !,
    backtrack(Disj,PatOut,Result).
eval([(G1;G2)|Conj],PatIn,Disj,PatOut,Result) :- !,
    copy_term(alt(PatIn, conj([G2|Conj])), Branch),
    disjoin(Branch,Disj,NewDisj),
    eval([G1|Conj],PatIn,NewDisj,PatOut,Result).
eval([conj(Cs)|Conj],PatIn,Disj,PatOut,Result) :- !,
    append(Cs,Conj,NewConj),
    eval(NewConj,PatIn,Disj,PatOut,Result).
eval([shift(Term)|Conj],PatIn,Disj,PatOut,Result) :- !,
    PatOut = PatIn,
    Disj = alt(BranchPatIn,Branch),
    Result = shift(Term, conj(Conj),BranchPatIn,Branch).
eval([reset(RPattern,RGoal,RResult)|Conj],PatIn,Disj,PatOut,Result):- !,
    copy_term(RPattern-RGoal,RPatIn-RGoalCopy),
    empty_alt(RDisj),
    eval([RGoalCopy],RPatIn,RDisj,RPatOut,RResultFresh),
    eval([RPattern=RPatOut,RResult=RResultFresh|Conj]
        ,PatIn,Disj,PatOut,Result).
eval([Call|Conj],PatIn,Disj,PatOut,Result) :- !,
    findall(Call-Body,clause(Call,Body), Clauses),
    ( Clauses = [] -> backtrack(Disj,PatOut,Result)
    ; disjoin_clauses(Call,Clauses,ClausesDisj),
        eval([ClausesDisj|Conj],PatIn,Disj,PatOut,Result)
    ).
```

Fig. 1. Meta-Interpreter Core

If there is no alternative branch, it sets the Result to failure. Otherwise, it resumes with the alternative branch. Note that by managing its own backtracking, eval/5 is entirely deterministic with respect to the meta-level Prolog system.

When the first conjunct is a disjunction (G1;G2) (11-14), the meta-interpreter adds (a renamed apart copy of) (G2, Conj) to the alternative branches with disjoin/3 and proceeds with [G1|Conj].

```
disjoin(alt(_,fail),Disjunction,Disjunction) :- !.
disjoin(Disjunction,alt(_,fail),Disjunction) :- !.
```

```
disjoin(alt(P1,G1),alt(P2,G2),Disjunction) :-
    Disjunction = alt(P3, (P1 = P3, G1 ; P2 = P3, G2)).
```

Note that we have introduced a custom built-in conj (Conj) that turns a list of goals into an actual conjunction. It is handled (15-17) by prepending the goals to the current list of conjuncts, and never actually builds the explicit conjunction.

When the first goal is shift (Term) (18-21), this is handled similarly to an empty conjunction, except that the result is a shift/4 term which contains Term and the remainder of the conjunction in addition the branch information.

When the first goal is a reset (RPattern, RGoal, RResult) (22-27), the metainterpreter sets up an isolated call to eval/5 for this goal. When the call returns, the meta-interpreter passes on the results and resumes the current conjunction Conj. Notice that we are careful that this does not result in meta-level failure by meta-interpreting the unification.

Finally, when the first goal is a call to a user-defined predicate (28-33), the meta-interpreter collects the bodies of the predicate's clauses whose head unifies with the call. If there are none, it backtracks explicitly. Otherwise, it builds an explicit disjunction with disjoin_clauses, which it pushes on the conjunction stack.

```
disjoin_clauses(_G,[],fail) :- !.
disjoin_clauses(G, [GC-Clause], (G=GC,Clause)) :- !.
disjoin_clauses(G,[GC-Clause|Clauses], ((G=GC,Clause) ; Disj)) :-
    disjoin_clauses(G,Clauses,Disj).
```

An example execution trace of the interpreter can be found in $\mathbb{C}$

Toplevel The toplevel(Goal)-predicate initialises the core interpreter with a conjunction containing only the given goal, the pattern and pattern copy set to (distinct) copies of the goal, and an empty disjunction. It interprets the result by non-deterministically producing all the answers to Goal and signalling an error for any unhandled shift/1.

```
toplevel(Goal) :-
    copy_term(Goal,GoalCopy),
    PatIn = GoalCopy,
    empty_alt(Disj),
    eval([GoalCopy],PatIn,Disj,PatOut,Result),
    ( Result = success(BranchPatIn,Branch) ->
        ( Goal = PatOut ; Goal = BranchPatIn, toplevel(Branch))
    ; Result = shift(_,_,_,_) ->
        write('toplevel: uncaught shift/1.\n'), fail
    ; Result = failure ->
        fail
    ).
```

```
bound(V) :- shift(V).
bb(Value,Data,Goal,Min) :-
    reset(Data,Goal,Result),
    bb_result(Result,Value,Data,Min).
bb_result(success(BranchCopy,Branch),Value,Data,Min) :-
    ( Data @< Value -> bb(Data,BranchCopy,Branch,Min)
    ; bb(Value,BranchCopy,Branch,Min)
    ).
bb_result(shift(ShiftTerm,Cont,BranchCopy,Branch),Value,Data,Min) :-
    ( ShiftTerm @< Value ->
        bb(Value,Data,(Cont ; (BranchCopy = Data,Branch)),Min)
    ; bb(Value,BranchCopy,Branch,Min)
    ).
bb_result(failure,Value,_Data,Min) :- Value = Min.
```

Fig. 2. Branch-and-Bound Effect Handler.

## 4 Case Studies

To illustrate the usefulness and practicality of our approach, we present two case studies that use the new reset/3 and shift/1.

### 4.1 Branch-and-Bound: Nearest Neighbour Search

Branch-and-bound is a well-known general optimisation strategy, where the solutions in certain areas or branches of the search space are known to be bounded. Such branches can be pruned, when their bound does not improve upon a previously found solution, eliminating large swaths of the search space in a single stroke.

We provide an implementation of branch-and-bound (see Figure 2) that is generic, i.e., it is not specialised for any application. In particular it is not specific to nearest neighbour search, the problem on which we demonstrate the branch-and-bound approach here.

The framework requires minimal instrumentation: it suffices to begin every prunable branch with bound $(\mathrm{V})$, where V is a lower bound on the values in the branch $\sqrt[3]{ }$

1. If the Goal succeeds normally (i.e., Result is success), then Data contains a new solution, which is only accepted if it is an improvement over the existing Value. The handler then tries the next Branch.
2. If the Goal calls bound $(\mathrm{V}), \mathrm{V}$ is compared to the current best Value:
[^0]```
nn((X,Y),BSP,D-(NX,NY)) :-
    ( BSP = xsplit((SX,SY),Left,Right) ->
            DX is X - SX,
            branch((X,Y), (SX,SY), Left, Right, DX, D-(NX,NY))
    ; BSP = ysplit((SX,SY),Up,Down) ->
            DY is Y - SY,
            branch((X,Y), (SX,SY), Up, Down, DY, D-(NX,NY))
    ).
branch((X,Y), (SX,SY), BSP1, BSP2, D, Dist-(NX,NY)) :-
    ( D < O >> % Find out which partition contains (X,Y).
            TargetPart = BSP1, OtherPart = BSP2, BoundaryDistance is -D
    ;
        TargetPart = BSP2, OtherPart = BSP1, BoundaryDistance is D
    ),
    ( nn((X,Y), TargetPart, Dist-(NX,NY))
    ; Dist is (X - SX) * (X - SX) + (Y - SY) * (Y - SY),
        (NX,NY) = (SX,SY)
    ; bound(BoundaryDistance-nil),
        nn((X,Y), OtherPart,Dist-(NX,NY))
    ).
run_nn((XO,YO),BSP, (NX,NY)) :-
    toplevel(bb(10-nil,D-(X,Y),nn((XO,YO),BSP,D-(X,Y)),_-(NX,NY))).
```

Fig. 3. 2D Nearest Neighbour Search with Branch-and-Bound.

- if it is less than the current value, then Cont could produce a solution that improves upon the current value, and thus must be explored. The alternative Branch is disjoined to Cont, and DataCopy is restored to Data (ensuring that a future reset/3 copies the right variables);
- if it is larger than or equal to the current value, then Cont can be safely discarded.

3. Finally, if the goal fails entirely, Min is the current minimum Value.

Nearest Neighbour Search The code in Figure 3 shows how the branch and bound framework efficiently solves the problem of finding the point (in a given set) that is nearest to a given target point on the Euclidean plane.

The run_nn/3 predicate takes a point (X,Y), a Binary Space Partitioning (BSP)-tred that represents the set of points, and returns the point, nearest to ( $\mathrm{X}, \mathrm{Y}$ ). The algorithm implemented by $\mathrm{nn} / 3$ recursively descends the BSP-tree. At each node it first tries the partition to which the target point belongs, then the point in the node, and finally the other partition. For this final step we can give an easy lower bound: any point in the other partition must be at least as

[^1]

Fig. 4. Nearest-Neighbour Search using a BSP-tree
far away as the (perpendicular) distance from the given point to the partition boundary.

As an example, we search for the point nearest to $(1,0.1)$ in the set $\{(0.5,0.5)$, $(0,0),(-0.5,0),(-0.75,-0.5)\}$. Figure 4 shows a BSP-tree containing these points, the solid lines demarcate the partitions. The algorithm visits the points $(0.5,0.5)$ and $(0,0)$, in that order. The shaded area is never visited, since the distance from $(1,0.1)$ to the vertical boundary through $(0,0)$ is greater than the distance to $(0.5,0.5)$ ( 1 and about 0.64 ). The corresponding call to run_nn/3 is:

```
?- BSP = xsplit((0,0),
        ysplit((-0.5,0),leaf,xsplit((-0.75,-0.5),leaf,leaf)),
        ysplit((0.5,0.5),leaf,leaf)),
    run_nn((1,0.1),BSP,(NX,NY)).
NX = NY, NY = 0.5.
```


### 4.2 Probabilistic Programming

Probabilistic programming languages (PPLs) are programming languages designed for probabilistic modelling. In a probabilistic model, components behave in a variety of ways-just like in a non-deterministic model-but do so with a certain probability.

Instead of a single deterministic value, the execution of a probabilistic program results in a probability distribution of a set of values. This result is produced by probabilistic inference 156 , for which there are many strategies and algorithms, the discussion of which is out of scope here. Here, we focus on one concrete probabilistic logic programming languages: PRISM [9.

A PRISM program consists of Horn clauses, and in fact, looks just like a regular Prolog program. However, we distinguish two special predicates:

- values_x (Switch, Values, Probabilities) This predicate defines a probabilistic switch Switch, that can assume a value from Values with the probability that is given at the corresponding position in Probabilities (the contents of Probabilities should sum to one).

```
msw(Key,Value) :- shift(msw(Key,Value)).
prob(Goal) :-
    prob(Goal,ProbOut),
    write(Goal), write(': '), write(ProbOut), write('\n').
prob(Goal,ProbOut) :-
    copy_term(Goal,GoalCopy),
    reset(GoalCopy,GoalCopy,Result),
    analyze_prob(GoalCopy,Result,ProbOut).
analyze_prob(_,failure,0.0).
analyze_prob(_,success(_,_),1.0).
analyze_prob(_,shift(msw(K,V),C,_,Branch),ProbOut) :-
    values_x(K,Values,Probabilities),
    msw_prob(V,C,Values,Probabilities,0.0,ProbOfMsw),
    prob(Branch,BranchProb),
    ProbOut is ProbOfMsw + BranchProb.
```

Fig. 5. An implementation of probabilistic programming with delimited control.

- msw(Switch,Value) This predicate samples a value Value from a switch Switch. For instance, if the program contains a switch declared as values_x ( coin, [h, t], [0.4, 0.6]), then msw(coin, V) assigns $h$ (for heads) to $V$ with probability 0.4 , and t (for tails) with probability 0.6 . Remark that each distinct call to msw leads to a different sample from that switch. For instance, in the query msw (coin, X ), msw ( $\operatorname{coin}, \mathrm{Y}$ ), the outcome could be either $(h, h),(t, t),(h, t)$ or $(t, h)$.

Consider the following PRISM program, the running example for this section:

```
values_x(coin1,[h,t],[0.5,0.5]).
values_x(coin2,[h,t],[0.4,0.6]).
twoheads :- msw(coin1,h),msw(coin2,h).
onehead :- msw(coin1,V), (V = t, msw(coin2,h) ; V = h).
```

This example defines two predicates: twoheads which is true if both coins are heads, and onehead which is true if either coin is heads. However, note the special structure of onehead: PRISM requires the exclusiveness condition, that is, branches of a disjunction cannot be both satisfied at the same time. The simpler goal msw (coin1,heads) ; msw (coin2, heads) violates this assumption.

The code in Figure5interprets this program. Line 1 defines msw/2 as a simple shift. Lines 6-9 install a reset/3 call over the goal, and analyse the result. The result is analysed in the remaining lines: A failure never succeeds, and thus has success probability 0.0 (line 9). Conversely, a successful computation has a success probability of 1.0 (line 10). Finally, the probability of a switch (lines 11-15) is the sum of the probability of the remainder of the program given each possible value of the switch multiplied with the probability of that value, and summed with the probability of the alternative branch.

The predicate msw_prob finds the joint probability of all choices. It iterates over the list of values, and sums the probability of their continuations.

```
msw_prob(_,_, [], [],Acc,Acc).
msw_prob(V,C,[Value|Values], [Prob|Probs],Acc,ProbOfMsw) :-
    prob((V = Value,C),ProbOut),
    msw_prob(V,C,Values,Probs,Prob*ProbOut + Acc,ProbOfMsw).
```

Now, we can compute the probabilities of the two predicates above:

```
?- toplevel(prob(twoheads)).
twoheads: 0.25
?- toplevel(prob(onehead)).
onehead: 0.75
```

In Appendix B. 3 we show how to implement the semantics a definite, non-looping fragment of ProbLog [6], another logic PPL, on top of the code in this section.

## 5 Properties of the Meta-Interpreter

In this section we establish two important correctness properties of our metainterpreter. The proofs of these properties are in the corresponding Appendices A. 1 and A. 2 The first theorem establishes the soundness of the meta-interpreter, i.e., if a program (not containing shift/1 or reset/3) evaluates to success, then an SLD-derivation of the same answer must exist.

Theorem 1 (Soundness) For all lists of goals $\left[A_{1}, \ldots, A_{n}\right]$, terms $\alpha, \beta, \gamma, \nu$, variables $P, R$ conjunctions $B_{1}, \ldots, B_{m} ; C_{1}, \ldots, C_{k}$ and substitutions $\theta$, if

$$
\begin{aligned}
& ?-\operatorname{eval}\left(\left[A_{1}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma, C_{1}, \ldots, C_{k}\right)
\end{aligned}
$$

and the program contains neither shift/1 nor reset/3, then SLD-resolution 5 finds the following derivation:

$$
\leftarrow\left(A_{1}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right)
$$

(with solution $\theta$ s.t. $\alpha \theta=\nu$ )
Conversely, we want to argue that the meta-interpreter is complete, i.e., if SLD-derivation finds a refutation, then meta-interpretation-provided that it terminates - must find the same answer eventually. The theorem is complicated

[^2]somewhat by the fact that the first answer that the meta interpreter arrives at might not be the desired one due to the order of the clauses in the program. To deal with this problem, we use the operator ? ${ }_{-}$p, which is like ?-, but allows a different permutation of the program in every step.

Theorem 2 (Completeness) For all goals $\leftarrow A_{1}, \ldots, A_{n}$, if has solutiono $\theta$, then

$$
\begin{aligned}
& ?-\mathrm{p} \operatorname{eval}\left(\left[A_{1}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right), R=\alpha \theta .
\end{aligned}
$$

Together, these two theorems show that our meta-interpreter is a conservative extension of the conventional Prolog semantics.

## 6 Related Work

Conjunctive Delimited Control Disjunctive delimited control is the culmination of a line of research on mechanisms to modify Prolog's control flow and search, which started with the hook-based approach of Tor [11] and was followed by the development of conjunctive delimited control for Prolog [1012].

The listing below shows that disjunctive delimited control entirely subsumes conjunctive delimited control. The latter behaviour is recovered by disjoining the captured disjunctive branch. We believe that TOR is similarly superseded.

```
nd_reset(Goal,Ball,Cont) :-
    copy_term(Goal,GoalCopy),
    reset(GoalCopy,GoalCopy,R),
    ( R = failure -> fail
    ; R = success(BranchPattern,Branch) ->
        ( Goal = GoalCopy, Cont = 0
        ; Goal = BranchPattern, nd_reset(Branch,Ball,Cont))
    ; R = shift(X,C,BranchPattern,Branch) ->
            ( Goal = GoalCopy, Ball = X, Cont = C
            ; Goal = BranchPattern, nd_reset(Branch,Ball,Cont))
    ).
```

Abdallah [1] presents a higher-level interface for (conjunctive) delimited control on top of that of Schrijvers et al. [10]. In particular, it features prompts, first conceived in a Haskell implementation by Dyvbig et al. [4], which allow shifts to dynamically specify up to what reset to capture the continuation. We believe that it is not difficult to add a similar prompt mechanism on top of our disjunctive version of delimited control.

Interoperable Engines Tarau and Majumdar's Interoperable Engines [14] propose engines as a means for co-operative coroutines in Prolog. An engine is an independent instance of a Prolog interpreter that provides answers to the main interpreter on request.

```
get(Interactor,Answer) :-
    get_engine(Interactor,Engine), % get engine state
    run_engine(Engine,NewEngine,Answer), % run up to the next answer
    update_engine(Interactor,NewEngine). % store the new engine state
return(X) :- shift(return(X)).
run_engine(engine(Pattern,Goal),NewEngine,Answer) :-
    reset(Pattern,Goal, Result),
    run_engine_result(Pattern,NewEngine,Answer,Result).
run_engine_result(Pattern,NewEngine,Answer,failure) :-
    NewEngine = engine(Pattern,fail),
    Answer = no.
run_engine_result(Pattern,NewEngine,Answer,success(BPattern,B)) :-
    NewEngine = engine(BPattern,B),
    Answer = the(Pattern).
run_engine_result(Pattern,NewEngine,Answer,S) :-
    S = shift(return(X),C,BPattern,B)
    BPattern = Pattern,
    NewEngine = engine(Pattern,(C;B)),
    Answer = the(X).
```

Fig. 6. Interoperable Engines in terms of delimited control.

The predicate new_engine(Pattern, Goal, Interactor) creates a new engine with answer pattern Pattern that will execute Goal and is identified by Interactor. The predicate get (Interactor, Answer) has an engine execute its goal until it produces an answer (either by proving the Goal, or explicitly with return/1). After this predicate returns, more answers can be requested, by calling get/2 again with the same engine identifier. The full interface also allows bi-directional communication between engines, but that is out of scope here.

Figure 6 shows that we can implement the get/2 engine interface in terms of delimited control (the full code is available in Appendix B.2). The opposite, implementing disjunctive delimited control with engines, seems impossible as engines do not provide explicit control over the disjunctive continuation. Indeed, get/2 can only follow Prolog's natural left-to-right control flow and thus we cannot, e.g., run the disjunctive continuation before the conjunctive continuation, which is trivial with disjunctive delimited control.

Tabling without non-bactrackable variables Tabling [138] is a well-known technique that eliminates the sensitivity of SLD-resolution to clause and goal ordering, allowing a larger class of programs to terminate. As a bonus, it may improve the run-time performance (at the expense of increased memory consumption).

One way to implement tabling-with minimal engineering impact to the Prolog engine - is the tabling-as-a-library approach proposed by Desouter et al. [3. This approach requires (global) mutable variables that are not erased by backtracking to store their data structures in a persistent manner. With the new
reset/3 predicate, this is no longer needed, as (non-backtracking) state can be implemented in directly with disjunctive delimited control.

## 7 Conclusion and Future Work

We have presented disjunctive delimited control, an extension to delimited control that takes Prolog's non-deterministic nature into account. This is a conservative extension that enables implementing disjunction-related language features and extensions as a library.

In future work, we plan to explore a WAM-level implementation of disjunctive delimited control, inspired by the stack freezing functionality of tabling engines, to gain access to the disjunctive continuations efficiently. Similarily, the use of copy_term/2 necessitated by the current API has a detrimental impact on performance, which might be overcome by a sharing or shallow copying scheme.

## References

1. Abdallah, S.: More declarative tabling in Prolog using multi-prompt delimited control. CoRR abs/1708.07081 (2017)
2. Danvy, O., Filinski, A.: Abstracting control. pp. 151-160. LFP '90 (1990)
3. Desouter, B., van Dooren, M., Schrijvers, T.: Tabling as a library with delimited control. TPLP 15(4-5), 419-433 (2015)
4. Dyvbig, R.K., Jones, S.P., Sabry, A.: A monadic framework for delimited continuations. Tech. Rep. 615, Computer Science Department Indiana University (2005)
5. Felleisen, M.: The theory and practice of first-class prompts. pp. 180-190. POPL '88 (1988)
6. Fierens, D., den Broeck, G.V., Renkens, J., Shterionov, D.S., Gutmann, B., Thon, I., Janssens, G., Raedt, L.D.: Inference and learning in probabilistic logic programs using weighted boolean formulas. TPLP 15(3), 358-401 (2015)
7. Saleh, A.H., Schrijvers, T.: Efficient algebraic effect handlers for Prolog. TPLP 16(5-6), 884-898 (2016)
8. Santos Costa, V., Rocha, R., Damas, L.: The YAP Prolog system. TPLP 12(1-2), 5-34 (2012)
9. Sato, T.: Generative modeling by PRISM. In: ICLP. LNCS, vol. 5649, pp. 24-35. Springer (2009)
10. Schrijvers, T., Demoen, B., Desouter, B., Wielemaker, J.: Delimited continuations for Prolog. TPLP 13(4-5), 533-546 (2013)
11. Schrijvers, T., Demoen, B., Triska, M., Desouter, B.: Tor: Modular search with hookable disjunction. Sci. Comput. Program. 84, 101-120 (2014)
12. Schrijvers, T., Wu, N., Desouter, B., Demoen, B.: Heuristics entwined with handlers combined: From functional specification to logic programming implementation. In: Proceedings of PPDP 2014. pp. 259-270. ACM (2014)
13. Swift, T., Warren, D.S.: XSB: Extending Prolog with tabled logic programming. TPLP 12(1-2), 157-187 (Jan 2012)
14. Tarau, P., Majumdar, A.K.: Interoperating logic engines. In: Gill, A., Swift, T. (eds.) PADL 2009 Proceedings. LNCS, vol. 5418, pp. 137-151. Springer (2009)
15. Wood, F.D., van de Meent, J., Mansinghka, V.: A new approach to probabilistic programming inference. In: AISTATS. JMLR Workshop and Conference Proceedings, vol. 33, pp. 1024-1032. JMLR.org (2014)

## Appendix Table of Contents

A Correctness Proofs ..... 17
A. 1 Evaluation Is Sound ..... 17
A. 2 Evaluation Is Complete ..... 23
B Additional Examples ..... 24
B. 1 Negation ..... 24
B. 2 Interoperable Engines ..... 24
B. 3 ProbLog ..... 25
C Example Meta-Interpreter Trace ..... 27
D Full Meta-Interpreter ..... 29

## A Correctness Proofs

## A. 1 Evaluation Is Sound

Theorem 1 (Soundness) For all lists of goals $\left[A_{1}, \ldots, A_{n}\right]$, terms $\alpha, \beta, \gamma, \nu$, variables $P, R$ conjunctions $B_{1}, \ldots, B_{m} ; C_{1}, \ldots, C_{k}$ and substitutions $\theta$, if

$$
\begin{aligned}
& ?-\operatorname{eval}\left(\left[A_{1}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma, C_{1}, \ldots, C_{k}\right)
\end{aligned}
$$

and the program contains neither shift/1 nor reset/3, then SLD-resolution ${ }^{6}$ to finds the following derivation:

$$
\begin{gathered}
\leftarrow\left(A_{1}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right) \\
\vdots \\
\square \\
\text { (with solution } \theta \text { s.t. } \alpha \theta=\nu)
\end{gathered}
$$

Proof. By induction on the Prolog derivation.
Case [] In this case the only possible result is $P=\alpha$ :

$$
\begin{aligned}
& \text { ?- } \operatorname{eval}\left([], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\alpha, R=\operatorname{success}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right) .
\end{aligned} \quad \begin{gathered}
\leftarrow \operatorname{true} ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right) \\
\leftarrow \operatorname{true}(=\square) \\
\text { (with solution } \epsilon \text { ) }
\end{gathered}
$$

where $\epsilon$ is the empty substitution.
Case [true/ Conj] In this case the body contains only a direct recursive call, we have:

$$
\begin{aligned}
& \text { ?- } \operatorname{eval}\left(\left[\text { true }, A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) \text {. } \\
& P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right) \text {. } \\
& \Longleftrightarrow \quad \begin{array}{l}
\text { ?- } \operatorname{eval}\left(\left[A_{2} \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right) .
\end{array} \\
& \leftarrow\left(A_{2}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right) \\
& \Longrightarrow \quad \vdots \\
& \text { (with solution } \theta \text { s.t. } \alpha \theta=\nu \text { ) } \\
& \leftarrow\left(\text { true }, A_{2}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right) \\
& \Longrightarrow \\
& \text { (with solution } \theta \text { s.t. } \alpha \theta=\nu \text { ) }
\end{aligned}
$$

| Definition and Deter- <br> minism of eval $/ 5$ |
| :--- | minism of eval/5

Induction

SLD-resolution

[^3]Case [(G1,G2)/ConJ] A straightforward calculation gives the desired result:

> ?- $\operatorname{eval}\left(\left[\left(G_{1}, G_{2}\right), A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right)$.
> $P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right)$.

## Definition and Determinism of eval/5

?- $\operatorname{eval}\left(\left[G_{1}, G_{2}, A_{2} \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right)$.
$\Longleftrightarrow P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right)$.
$\leftarrow\left(G_{1}, G_{2}, A_{2}, \ldots, A_{n}\right.$, true $) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right)$

| $\Longrightarrow$ | $\vdots$ | Induction |
| :---: | :---: | ---: |
|  | $\square$ <br> (with solution $\theta$ s.t. $\alpha \theta=\nu)$ |  |
|  | $\leftarrow\left(\left(G_{1}, G_{2}\right), A_{2}, \ldots, A_{n}\right.$, true $) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right)$ |  |
|  | $\vdots$ |  |
|  | $\square$ | Associativity |

(with solution $\theta$ s.t. $\alpha \theta=\nu$ )

Case [fail/ConJ] This case immediately calls backtrack/3:

$$
\begin{aligned}
& ?-\operatorname{eval}\left(\left[f \operatorname{fail}, A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right) . \\
& \Longleftrightarrow ?-\operatorname{backtrack}\left(\operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right) .
\end{aligned}
$$

| $\begin{array}{l}\text { Definition and Deter- } \\ \text { minism of eval } / 5\end{array}$ |
| :--- |

At this point, we can see that $\operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right)$ cannot be empty, for otherwise $R=$ failure. Execution of the else branch then gives:


Case $[(G 1 ; G 2), \operatorname{Cons}]$ Assume that the result of the copy_term/2 call is alt $\left(\delta, \operatorname{conj}\left(\left[D_{1}, \ldots, D_{n}\right]\right)\right)$, where all the variables are fresh, such that

$$
\begin{equation*}
\delta \sigma=\alpha \sigma \Rightarrow\left(D_{1}, \ldots, D_{n}\right) \sigma=\left(G_{2}, A_{2}, \ldots, A_{n}\right) \sigma \tag{1}
\end{equation*}
$$

Then the result of disjoining this with $\operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right)$ is the new disjunction

$$
\xi=\operatorname{alt}\left(F,\left(\left(\delta=F, \operatorname{conj}\left(\left[D_{1}, \ldots, D_{n}\right]\right)\right) ;\left(\beta=F, B_{1}, \ldots, B_{m}\right)\right)\right)
$$

with $F$ fresh. Now we reason as follows:

$$
\begin{aligned}
& ?-\operatorname{eval}\left(\left[\left(G_{1} ; G_{2}\right) A_{2}, \ldots, A_{n}\right], \alpha, \text { alt }\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right) . \\
\Longleftrightarrow & ?-\operatorname{eval}\left(\left[G_{1}, A_{2}, \ldots, A_{n}\right], \alpha, \xi, P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right) . \\
& \text { Definition and Determinism of eval } / 5 \\
& \leftarrow\left(G_{1}, A_{2}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=F,\left(\left(\delta=F, \operatorname{conj}\left(\left[D_{1}, \ldots, D_{n}\right]\right)\right) ;\left(\beta=F, B_{1}, \ldots, B_{m}\right)\right)\right) \\
\vdots & \vdots \\
& \square \\
& \quad \text { (with solution } \theta \text { s.t. } \alpha \theta=\nu)
\end{aligned}
$$

Associativity, Distributivity, resolution of conj/1, $F$ is local variable
$\leftarrow\left(G_{1}, A_{2}, \ldots, A_{n}\right.$, true $) ;\left(\alpha=\delta, D_{1}, \ldots, D_{n}\right) ;\left(\alpha, \beta, B_{1}, \ldots, B_{m}\right)$


$$
\text { (with solution } \theta \text { s.t. } \alpha \theta=\nu \text { ) }
$$

Note that if $\sigma$ refutes $\leftarrow \alpha=\delta, D_{1}, \ldots, D_{n}$; then $\delta \sigma=\alpha \sigma$, hence (11) $\left(D_{1}, \ldots, D_{n}\right) \sigma=$ $\left(G_{2}, A_{2}, \ldots, A_{n}\right) \sigma$. Thus, $\sigma$ also refutes $\leftarrow G_{2}, A_{2}, \ldots, A_{n}$.

So, we may write the above as:
 (with solution $\theta$ s.t. $\alpha \theta=\nu$ )

## Distributivity

$\leftarrow\left(\left(G_{1} ; G_{2}\right), A_{2}, \ldots, A_{n}\right.$, true $) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right)$

## $\Longrightarrow$

(with solution $\theta$ s.t. $\alpha \theta=\nu$ )

## Case [conj(Cs)/ConJ]

$$
?-\operatorname{eval}\left(\left[\operatorname{conj}\left(\left[D_{1}, \ldots, D_{l}\right]\right), A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right)
$$

$P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right)$.

$$
\begin{aligned}
& \text { Definition and Determinism of eval/5 } \\
& \text { ?- } \operatorname{eval}\left(\left[D_{1}, \ldots, D_{l}, A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right)
\end{aligned}
$$

## Induction



## Induction



Case [C/ConJ] In this case there are two possibilities: if no clauses match with the head, Clauses is empty, and the procedure backtracks. In this case the reasoning is identical to the case [faillConJ].

Otherwise, assume we have a list $\left[(H:-B o d y)_{1}, \ldots,(H:-B o d y)_{l}\right]$ of clauses that match $C$. After disjoin_clauses $/ 3$, this list becomes a disjunction $(C=$ $H_{1}$, Body $\left._{1}\right) ; \cdots ;\left(C=H_{l}\right.$, Body $\left._{l}\right)$.

From the recursive call we get:

$$
\begin{aligned}
& \text { ?- } \operatorname{eval}\left(\left[C \mid A_{2}, \cdots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\nu, R=\operatorname{result}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right)
\end{aligned}
$$

## Definition and Determinism of eval/5

```
    ?- \(\operatorname{eval}\left(\left[\left(C=H_{1}, \operatorname{Bod} y_{1}\right) ; \cdots ;\left(C=H_{l}, B o d y_{l}\right) \mid A_{2}, \cdots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right)\).
\(\Longleftrightarrow P=\nu, R=\operatorname{result}\left(\gamma\left(C_{1}, \ldots, C_{k}\right)\right)\)
```


## Induction

$$
\left.\leftarrow\left(\left(C=H_{1}, \text { Body }_{1}\right) ; \cdots ;\left(C=H_{l}, \text { Body } y_{l}\right)\right), A_{2}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right)
$$


(with solution $\theta$, s.t. $\alpha \theta=\nu$ )

If $\leftarrow\left(\left(C=H_{1}, \operatorname{Bod} y_{1}\right) ; \cdots ;\left(C=H_{l}, B o d y_{l}\right)\right), A_{2}, \ldots, A_{n}$, true $)$ has a refutation $\theta$, then let $\theta=\theta_{i} \theta^{\prime}$ such that $\theta_{i}$ refutes $\leftarrow C=H_{i}, B o d y_{i}$ and $\theta^{\prime}$ refutes
$\leftarrow A_{2}, \ldots, A_{n}:$

| $\leftarrow C=H_{i}$, Body $y_{i}$ | $\leftarrow \operatorname{Bod} y_{i} \sigma$ |  | $\leftarrow C$ |
| :---: | :---: | :---: | :---: |
| $\vdots$ | $\vdots$ | $\vdots$ | $\leftarrow \operatorname{Bod} y_{i} \sigma$ |
| $\square$ | s.t. $C \sigma=H_{i} \sigma \Longrightarrow$ | $\vdots$ |  |
| $\left(\right.$ with solution $\left.\theta_{i}\right)$ |  | (with solution $\left.\theta_{i}^{\prime}\right)$ | $\square$ |

Then $\theta$ also refutes $\leftarrow C, A_{2}, \ldots, A_{n}$, so we can conclude:

$$
\leftarrow\left(C, A_{2}, \ldots, A_{n}, \text { true }\right) ;\left(\alpha=\beta, B_{1}, \ldots, B_{m}\right)
$$

(with solution $\theta$, s.t. $\alpha \theta=\nu$ )

## A. 2 Evaluation Is Complete

Theorem 2 For all goals $\leftarrow A_{1}, \ldots, A_{n}$, if

then

$$
\begin{aligned}
& ?-\mathrm{p} \operatorname{vval}\left(\left[A_{1}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right), R=\alpha \theta .
\end{aligned}
$$

where ${ }^{-}{ }_{-\mathrm{p}}$ is defined as ?-, but each step is allowed to use a different permutation of the program.

Proof. By induction on the length of the derivation.

Base $n=0: \leftarrow A_{1}, \ldots, A_{n}=\leftarrow$ true $=\square$ then $\theta$ is empty, and

$$
\begin{aligned}
& ?-\mathrm{p} \operatorname{eval}\left([], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) . \\
& P=\operatorname{success}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), R=\alpha
\end{aligned}
$$

by definition.
Induction Let

$$
\begin{aligned}
& \qquad \leftarrow A_{1}, A_{2}, \ldots, A_{n} \\
& \leftarrow G_{1}, \ldots, G_{f}, A_{2}, \ldots, A_{n} \theta_{1} \text { (clause } H \leftarrow G_{1}, \ldots, G_{f} ; A_{1} \theta_{1}=H \theta_{1} \text { ) } \\
& \quad \vdots \\
& \square \\
& \text { (with substitution } \theta=\theta_{1} \theta_{2} \text { ) }
\end{aligned}
$$

By induction we have,

$$
\begin{aligned}
& ?-\mathrm{p} \operatorname{eval}\left(\left[G_{1}, \ldots, G_{f}, A_{2}, \ldots, A_{n} \theta_{1}\right], \alpha \theta_{1}, \operatorname{alt}\left(\delta,\left(D_{1}, \ldots, D_{l}\right)\right), P, R\right) \\
& P=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right), S=\alpha \theta_{1} \theta_{2} . \\
& \Longleftrightarrow ?-\mathrm{p} \operatorname{eval}\left(\left[\left(A_{1}=H, G_{1}, \ldots, G_{f} ; \ldots\right), A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) \\
& P=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right), S=\alpha \theta_{1} \theta_{2} . \\
& \Longrightarrow ?-\mathrm{p} \operatorname{eval}\left(\left[A_{1}, A_{2}, \ldots, A_{n}\right], \alpha, \operatorname{alt}\left(\beta,\left(B_{1}, \ldots, B_{m}\right)\right), P, R\right) \\
& P=\operatorname{success}\left(\gamma,\left(C_{1}, \ldots, C_{k}\right)\right), S=\alpha \theta_{1} \theta_{2} .
\end{aligned}
$$

## B Additional Examples

## B. 1 Negation

```
not(G) :-
    copy_term(G,GC),
    reset(GC,GC,R),
    R = failure.
```


## B. 2 Interoperable Engines

```
engines(G) :-
```

    engines(G, []).
    new_engine(Pattern, Goal, Interactor) :-
shift(new_engine(Pattern, Goal, Interactor)).
get(Interactor, Answer) :-
shift(get(Interactor, Answer)).
return(X) :-
shift(return(X)).
engines(G,EngineList) :-
copy_term(G,GC),
reset (GC, GC, R),
engines_result (G,GC, EngineList, R).
engines_result(_, , _, failure) :-
fail.
engines_result(G,GC,EngineList,success(BC,B)) :-
( $G=G C$; $G=B C$, engines(B,EngineList)).
engines_result(G,GC,EngineList,S) :-
S = shift(new_engine(Pattern, Goal,Interactor), C, BC,B),
length(EngineList, Interactor),
copy_term(Pattern-Goal, PatternCopy-GoalCopy),
NewEngineList = [Interactor-engine(PatternCopy, GoalCopy)|EngineList],
$\mathrm{G}=\mathrm{GC}$,
$G=B C$,
engines ( ( $;$; B ), NewEngineList).
engines_result(G,GC, EngineList,S) :-
S = shift(get(Interactor, Answer), C, BC, B) ,
member(Interactor-Engine, EngineList),
run_engine(Engine, NewEngine, Answer),
update(Interactor, NewEngine, EngineList, NewEngineList),
G = GC,

```
    G = BC,
    engines((C;B),NewEngineList).
update(K,NewV,[K-_|T],[K-NewV|T]).
update(K,NewV,[OtherK-V|T],[OtherK-V|T2]) :-
    K \== OtherK,
    update(K,NewV,T,T2).
run_engine(engine(Pattern,Goal), NewEngine,Answer) :-
    reset(Pattern,Goal,Result),
    run_engine_result(Pattern,NewEngine,Answer,Result).
run_engine_result(Pattern,NewEngine,Answer,failure) :-
    NewEngine = engine(Pattern,fail),
    Answer = no.
run_engine_result(Pattern,NewEngine,Answer,success(BPattern,B)) :-
    NewEngine = engine(BPattern,B),
    Answer = the(Pattern).
run_engine_result(Pattern,NewEngine,Answer,shift(return(X),C,BPattern,B)) :-
    BPattern = Pattern,
    NewEngine = engine(Pattern,(C;B)),
    Answer = the(X).
```


## B. 3 ProbLog

```
fact(F) :-
    shift(fact(F,V)),
    V = t.
is_true(F,Pc) :- member(F-t,Pc).
is_false(F,Pc) :-member(F-f,Pc).
problog(Goal) :- problog(Goal,[]).
problog(Goal,Pc) :-
    reset(Goal,Goal,Result),
    analyze_problog(Result,Pc).
analyze_problog(success(_,_),_Pc).
analyze_problog(shift(fact(F,V),C,_,Branch),Pc) :-
    is_true(F,Pc),
    V = t,
    problog((C;Branch),Pc).
analyze_problog(shift(fact(F,V),C,_,Branch),Pc) :-
    is_false(F,Pc),
    V = f,
    problog((C;Branch),Pc).
```

```
analyze_problog(shift(fact(F,V),C,_,Branch),Pc) :-
    not(is_true(F,Pc)),
    not(is_false(F,Pc)),
    msw(F,V),
    problog((C;Branch),[F-V|Pc]).
analyze_problog(failure,_Pc) :- fail.
% NEGATION DOES NOT WORK.
```

Example usage:
\% 0.5 : : f1.
values_x(f1,[t,f],[0.5,0.5]).
f1 :- fact $(f 1)$.
\% 0.5 :: f2.
values_x (f2, [t,f],[0.5,0.5]).
f2 :- fact(f2).
p :- f1.
p :- f2
?- solutions(prob(problog((f1,f1)))).
problog((f1,f1)): 0.5
?- solutions(prob(problog(p))).
problog(p): 0.75

## C Example Meta-Interpreter Trace

Consider the simple program:
$\mathrm{p}(1)$.
$p(2):-\operatorname{shift}(2)$.

The following table shows the values of each of the arguments of the metainterpreter, while evaluating the goal $\mathrm{p}(\mathrm{X})$.

| PatIn Conj |  | Disj |  |
| :---: | :---: | :---: | :---: |
| X | [p(X)] | alt(Y,fail) |  |
|  | [ $\mathrm{X}=1$, true; | alt(Y,fail) |  |
| X | X = 2,shift(2))] |  |  |
| X | [(X=1, true) $]$ | alt (Z1, (Z1=X1, X1=2, shift (2) | ; Z1=Y,fail)) |
| X | [ $\mathrm{X}=1$, true] | alt (Z1, (Z1=X1, X1=2, shift (2) | ; Z1=Y,fail)) |
| 1 | [true] | alt (Z1, (Z1 = X1, X1 $=2$, shift (2) | ; Z1=Y,fail)) |
| 1 | [] | alt (Z1, (Z1=X1, X1=2, shift(2) | Z1=Y,fail)) |
| PatOut=1, Result=success(Z1, (Z1=X1, X1=2,shift(2) ; $\mathrm{Z} 1=\mathrm{Y}, \mathrm{fail}$ ) |  |  |  |

We can then evaluate the alternative branches:

| PatIn | Conj | Disj |
| :---: | :---: | :---: |
| Z1 | [(Z1=X1, X1=2,shift(2) ; $\mathrm{Z} 1=\mathrm{Y}, \mathrm{fail}) \mathrm{l}]$ | alt (A,fail) |
| Z1 | [(Z1=X1, X1 $=2, \operatorname{shift}(2))]$ | $\begin{gathered} \text { alt }(B,(Z 2=B, Z 2=Y, f a i l ; \\ A=B, f a i l) \end{gathered}$ |
| Z1 | [Z1=X1, X1=2,shift(2)] | $\begin{gathered} \text { alt }(B,(Z 2=B, Z 2=Y, f a i l ; \\ A=B, f a i l) \end{gathered}$ |
| X1 | [ $\mathrm{X} 1=2, \operatorname{shift}(2)]$ | $\begin{gathered} \text { alt }(B,(Z 2=B, Z 2=Y, f a i l ; \\ A=B, f a i l) \end{gathered}$ |
| 2 | [shift(2)) | $\begin{gathered} \text { alt }(\mathrm{B},(\mathrm{Z} 2=\mathrm{B}, \mathrm{Z2}=\mathrm{Y}, \mathrm{fail} ; \\ \mathrm{A}=\mathrm{B}, \mathrm{fail}) \end{gathered}$ |
|  | [] | $\begin{gathered} \text { alt }(\mathrm{B},(\mathrm{Z2}=\mathrm{B}, \mathrm{Z2}=\mathrm{Y}, \mathrm{fail} ; \\ \mathrm{A}=\mathrm{B}, \mathrm{fail}) \end{gathered}$ |
| Result $=$ shift ( $2, \operatorname{conj}([]), B,(Z 2=B, Z 2=Y, f a i l ; A=B, f a i l))$ |  |  |

and again

|  |  |
| :--- | :---: |
| $B[(Z 2=B, Z 2=Y, f a i l ; A=B, f a i l)]$ | alt (C,fail) |
| $\vdots \vdots$ | $\vdots$ |
| Y [fail] | alt(D,B1=D,A=B1,fail ; C=D,fail) |
| backtracking |  |
| D [(B1=D,A=B1,fail ; C=D,fail)] | alt(E,fail) |
| $\vdots \vdots$ | $\vdots$ |
| D [fail] |  |
| backtracking fails: Result=failure | alt(E,fail) |

## D Full Meta-Interpreter

The following code has been tested on SWI-Prolog versions 7.6 .3 and 8.0.3. It uses the type_check package to type-check the Prolog code. The type-checking annotations can be commented out, without loss of functionality on systems that do not support type-checking.

```
%%--------------------------------------------------------------------------------------
%% Meta-interpreter for disjunctive delimited control.
%%
%% The meta-interpreter supports most basic Prolog features, except if then
%% else.
%%
%% Author: Alexander Vandenbroucke (alexander.vandenbroucke@kuleuven.be)
%% Author: Tom Schrijvers (tom.schrijvers@kuleuven.be)
%%
%%--------------------------------------------------------------------------------
:- style_check(-singleton).
:- use_module(library(type_check)).
%% preliminaries
% The operator :=/2 is a more strictly typed synonym of =/2.
:- op(700,xfx,:=).
:- pred :=(A,A).
:=(X,Y) :- X = Y.
:- trust_pred copy_term(A,A).
:- trust_pred clause(pred,pred).
:- trust_pred shift(any).
:- trust_pred reset(P, pred,result(P)).
:- trust_pred write(any).
:- trust_pred length(list(A),integer).
:- trust_pred member(A,list(A)).
```

```
%%-------------------------------------------------------------------------------
%% all solutions
:- pred solutions(pred).
solutions(GO) :-
    copy_term(GO,G),
    copy_term(G,GC),
    empty_disj(EmptyDisj),
```

```
    do_reset((GC,true),G,GC,Result,EmptyDisj),
    analyze_solutions(GO,G,Result).
:- pred analyze_solutions(pred,pred,result(pred)).
analyze_solutions(GO,G,success(PatternCopy,Branch)) :-
    ( GO = G
    ; copy_term(PatternCopy,PC),
        empty_disj(EmptyDisj),
        do_reset((Branch,true),PC,PatternCopy,Result,EmptyDisj),
        analyze_solutions(G0,PC,Result)
    ).
analyze_solutions(_,_,shift(_,_,_,_)) :-
    write('solutions: unexpected shift/1.\n'),
    fail.
analyze_solutions(_,_,failure) :- fail.
```

```
%%--------------------------------------------------------------------------------------
```

%%--------------------------------------------------------------------------------------
%% Meta Interpreter
%% Meta Interpreter
- pred do_reset(pred,P,P,result(P),disjunction(P)).
- pred do_reset(pred,P,P,result(P),disjunction(P)).
%% do_reset(Conjuction,Pattern,PatternCopy,Result,Disjunction)
%% do_reset(Conjuction,Pattern,PatternCopy,Result,Disjunction)
%%
%%
%% * On success or shift, Pattern contains the current solution.
%% * On success or shift, Pattern contains the current solution.
%% Pattern MUST be an unbound variable.
%% Pattern MUST be an unbound variable.
%% * PatternCopy is a fresh copy of Pattern. Note: the contents of PatternCopy
%% * PatternCopy is a fresh copy of Pattern. Note: the contents of PatternCopy
%% are not always sensible after do_reset/5, hence PatternCopy should not be
%% are not always sensible after do_reset/5, hence PatternCopy should not be
%% inspected afterwards.
%% inspected afterwards.
%% * Afterwards Result is one of:
%% * Afterwards Result is one of:
%% - success(BranchPattern,Branch): BranchPattern is a fresh copy of
%% - success(BranchPattern,Branch): BranchPattern is a fresh copy of
%% Pattern, Branch is a disjunctive continuation (see below). In this case
%% Pattern, Branch is a disjunctive continuation (see below). In this case
%% PatternCopy and Pattern are unified, to carry out the current
%% PatternCopy and Pattern are unified, to carry out the current
%% substitution.
%% substitution.
%% - shift(Term,Conjunction,BranchPattern,Branch):
%% - shift(Term,Conjunction,BranchPattern,Branch):
%% Term is the term that was shifted, Conjuction is the conjunctive
%% Term is the term that was shifted, Conjuction is the conjunctive
%% continuation. BranchPattern is a fresh copy of Pattern, Branch is the
%% continuation. BranchPattern is a fresh copy of Pattern, Branch is the
%% disjunctive continuation. Pattern and PatternCopy are unified to carry
%% disjunctive continuation. Pattern and PatternCopy are unified to carry
%% out the current substitution.
%% out the current substitution.
%% - failure: when the conjunction and the disjunction fail
%% - failure: when the conjunction and the disjunction fail
%% Result MUST be an unbound variable.
%% Result MUST be an unbound variable.
%% * Conjunction: the current conjunctive goal. The variables it contains
%% * Conjunction: the current conjunctive goal. The variables it contains
%% should not occur in Pattern. Moreover, the final conjunct should always be
%% should not occur in Pattern. Moreover, the final conjunct should always be
%% true/0. The conjunction must NOT be empty.
%% true/0. The conjunction must NOT be empty.
%% * Disjunction: a disjunction/2, that contains the disjunctive goal (see

```
%% * Disjunction: a disjunction/2, that contains the disjunctive goal (see
```

```
%% empty_disj/1, disjoin/3 below.
%% single true/0-pattern (base case)
do_reset(true,Pattern,PatternCopy,Result,Disj) :-
    !,
    disjunction(BranchPattern,Branch) := Disj,
    Pattern := PatternCopy,
    Result := success(BranchPattern,Branch).
%% fail/0-pattern
do_reset((fail,_Conj),Pattern,_,Result,Disj) :-
    !,
    backtrack(Pattern,Result,Disj).
%% true/0-pattern
do_reset((true,Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    do_reset(Conj,Pattern,PatternCopy,Result,Disj).
%% conjunction pattern
do_reset(((G1,G2),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    do_reset((G1,(G2,Conj)),Pattern,PatternCopy,Result,Disj).
%% disjunction pattern
do_reset(((G1;G2),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    copy_term(disjunction(PatternCopy, (G2, Conj)), Branch),
    disjoin(Branch,Disj,NewDisj),
    do_reset((G1, Conj),Pattern,PatternCopy,Result,NewDisj).
%% shift/1-pattern
do_reset((shift(X),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    Pattern := PatternCopy,
    Disj := disjunction(BranchPattern,Branch),
    Result := shift(X,Conj,BranchPattern,Branch).
%% (new) reset/3-pattern
do_reset((reset(P,G,R), Conj),Pattern, PatternCopy,Result,Disj) :-
    !,
    copy_term(P-G,PC-GC),
    empty_disj(D),
    do_reset((GC,true),Q,PC,S,D),
    type_to_any(R,RAny),type_to_any(S,SAny),
```

```
    do_reset((P = Q, (RAny = SAny, Conj)),Pattern,PatternCopy,Result,Disj).
%% unification pattern
do_reset((X = Y,Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X = Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% unification pattern
do_reset((X = Y,Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X = Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%%term non-equivalence pattern
do_reset((X \== Y,Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X \== Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% copy_term/2-pattern
do_reset((copy_term(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( copy_term(X,Y) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% cut/0-pattern
do_reset((!,Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    do_reset(Conj,Pattern,PatternCopy,Result,Disj).
%% findall/3-pattern
do_reset((findall(T,G,L),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( findall(T,G,L) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% length/2-pattern
do_reset((length(L, X),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( length(L,X) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% member/2-pattern
do_reset((member(X,L),Conj),Pattern,PatternCopy,Result,Disj) :-
```

```
    !,
    ( member(X,L) -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% is/2-pattern
do_reset((is(R,Exp),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( R is Exp -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% (<)/2-pattern
do_reset((<(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X < Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% (>=)/2-pattern
do_reset((>=(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X >= Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% (@<)/2-pattern
do_reset((@< (X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X @< Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% (@>=)/2-pattern
do_reset((@>=(X,Y),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    ( X @>= Y -> do_reset(Conj,Pattern,PatternCopy,Result,Disj)
    ; backtrack(Pattern,Result,Disj)).
%% write/1-pattern
do_reset((write(T),Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    write(T),
    do_reset(Conj,Pattern,PatternCopy,Result,Disj).
%% clause pattern.
do_reset((G,Conj),Pattern,PatternCopy,Result,Disj) :-
    !,
    findall(GC-Body,(clause(G,Body), GC = G),Clauses),
    ( Clauses := [] ->
```

```
        backtrack(Pattern,Result,Disj)
    ; disjoin_clauses(G,Clauses,ClausesDisj),
        do_reset((ClausesDisj,Conj),Pattern,PatternCopy,Result,Disj)
    ).
:- pred disjoin_clauses(P,list(pair(P,pred)),pred).
disjoin_clauses(_,[],fail).
% The next clause is not necessary, but makes things prettier when tracing.
disjoin_clauses(G,[GC-Clause],(G=GC,Clause)) :- !.
disjoin_clauses(G,[GC-Clause|Clauses], ((G=GC,Clause) ; Disj) ) :-
    disjoin_clauses(G,Clauses,Disj).
:- pred backtrack(P,result(P),disjunction(P)).
%% backtracking
backtrack(Pattern,Result,Disj) :-
    ( empty_disj(Disj) ->
        Result := failure
    ; Disj := disjunction(PatternCopy,G) ->
        empty_disj(EmptyDisj),
        do_reset((G,true),Pattern,PatternCopy,Result,EmptyDisj)
        %% TODO: could be more efficient if we pattern match on G?
    ).
```

```
%%-
```

%%-
%% Disjunctions
%% Disjunctions
%%
%%
%% Disjunctions are of the from disjunction(Pattern,Goal), where Goal may
%% Disjunctions are of the from disjunction(Pattern,Goal), where Goal may
%% contain a disjunction.
%% contain a disjunction.
%% Disjunctions are neither commutative nor associative. However empty_disj/1
%% Disjunctions are neither commutative nor associative. However empty_disj/1
%% is still the unit for disjoin/3.
%% is still the unit for disjoin/3.
:- type disjunction(Pattern) ---> disjunction(Pattern,pred).
:- type disjunction(Pattern) ---> disjunction(Pattern,pred).
%% The empty disjunction.
%% The empty disjunction.
:- pred empty_disj(disjunction(Pattern)).
:- pred empty_disj(disjunction(Pattern)).
empty_disj(disjunction(_,fail)).
empty_disj(disjunction(_,fail)).
%% Disjoin two disjunctions.
%% Disjoin two disjunctions.
%%
%%
%% This operation is not commutative
%% This operation is not commutative
:- pred disjoin(disjunction(P),disjunction(P),disjunction(P)).
:- pred disjoin(disjunction(P),disjunction(P),disjunction(P)).
disjoin(disjunction(_,fail),Disjunction,Disjunction) :- !.
disjoin(disjunction(_,fail),Disjunction,Disjunction) :- !.
disjoin(Disjunction,disjunction(_,fail),Disjunction) :- !.

```
disjoin(Disjunction,disjunction(_,fail),Disjunction) :- !.
```

```
disjoin(disjunction(PC1,G1),disjunction(PC2,G2),Disjunction) :-
    PC1 := PC2,
    Disjunction := disjunction(PC1, (G1 ; G2)).
% NOTE: things could probably be made more efficient by not unifying PC1
% and PC2, and keeping the disjunctions explicit, s.t. no fresh copy of
% G2 needs to be made, but this is way simpler.
```

```
%%----------------------------------------------------------------------------------------
```

%%----------------------------------------------------------------------------------------
%% Results
%% Results
:- type result(R) ---> success(R,pred) ; shift(any,pred,R,pred) ; failure.

```
:- type result(R) ---> success(R,pred) ; shift(any,pred,R,pred) ; failure.
```


[^0]:    ${ }^{3}$ The framework searches for a minimal solution.

[^1]:    ${ }^{4}$ A BSP-tree is a tree that recursively partitions a set of points on the Euclidean plane, by picking points and alternately splitting the plane along the $x$ - or $y$-coordinate of those point. Splitting along the x-coordinate produces an xsplit/3 node, along the y-coordinate produces a ysplit/3 node.

[^2]:    ${ }^{5}$ Standard SLD-resolution, augmented with disjunctions and conj/1 goals.

[^3]:    ${ }^{6}$ Standard SLD-resolution, augmented with disjunctions and conj/1 goals.

